

TR-O-0031

39

空間回路網法を用いた
周波数特性解析プログラム

角田紀久夫

1990. 7. 27.

ATR光電波通信研究所

概要 (Abstract)

電磁界の数値解析手法はマイクロ波回路・アンテナ素子の設計において有力なツールとなっている。これらソフトウェアの充実を計るためATRでは各種解析プログラムの導入を進めている。

この度、空間回路網法の導入を図り、プログラムを作成した。この解析プログラムでは、3次元空間の時間ステップの電磁界解析を行い、過渡現象における電磁界の挙動をビジュアルに観測できるほか、フーリエ変換を用いて周波数特性を計算することができる。

この報告書は、プログラムコードに用いる基本計算式および本プログラムコードの使用方法を示すとともに、本手法を用いる上での注意事項を記している。

発行時 配付先 (Initial Distribution Specifications)

--

備考 (Notes)

--

目次

1.	まえがき	1
2	空間回路網法解析の概要	2
3	解析プログラムの使用法および プログラムコードの説明	10
4	プログラム使用例	34
5	まとめ	39

はじめに

近年、電子計算機の発達に伴い電磁界の時間軸解析手法の報告例が多くみうけられる。この種の解析法としては、空間回路網法^[1](以前はBergeron法と呼ばれていた)、差分時間領域法^[2](FD-TD法)、伝送線路行列法^[3](TLM法)がよく知られ、それぞれ日本、アメリカ、ヨーロッパで中心に研究されている。これらは共にマックスウエル方程式の差分解法であって、計算手順等が異なるものの、共に空間を離散化した格子点でモデル化し各格子点の電磁界の値を一離散時間前における電磁界値を用いて逐次求めている。

このような時間軸解析法の特徴は、電磁界の過渡(定常)状態をビジュアルに観察できるほか、パルス波入力に対する時間応答を計算できるため、フーリエ変換(FFT)によって広帯域な周波数特性を得られることである。

一方、マイクロ波受動回路の設計、例えば立体回路、平面回路(線路の不連続構成、多層構成など)やアンテナ等放射素子の設計においては電磁界解析ソフトは有力なツールである。現在ATRでは、有限要素法を立ち上げつつあるが、今のところ定常状態の解析を対象としており、回路網の広帯域な周波数特性を計算することには向いていない。そこで前記時間軸解析法の特徴に着目し、解析ツールとして空間回路網法を導入した。

以下、本レポートでは2章に本プログラムで用いる計算式を簡単に記し、3章にプログラムコードを記載した。4章は本プログラムの使用法を記すとともに、解析における注意点を記載した。本解析法に関する詳細は参考文献[1]を参照してほしい。

参考文献

- [1] 山下他：“電磁波問題の基礎解析法”、電子通信情報学会 昭62
- [2] A. Taflove, et al : “Numerical Solution of Steady-state Electromagnetic Scattering Problems Using the Time-Dependent Maxwell's Equations,” IEEE Trans. Microwave Theory & Tech., vol. 23,no.8,pp623-630, Aug. 1975
- [3] J.P. Hoefler : “ The Transmission-Line Matrix Method --- Theory and Applications ” IEEE Trans. Microwave Theory & Tech., vol. 33,no.10,pp882-893, Oct. 1985

2. 空間回路網法の概要

本章では、等価回路表現およびプログラムに用いる計算式について概略を記す。詳細については文献[1]-[3]を参照してほしい。

2.1 等価回路表現

本計算法では、空間は格子点でモデル化される。図2-1に解析に用いる基本格子点配置を示す。格子点はA-Fで示した6種類に分類され、X,Y,Z各方向には間隔 Δd で配置される。(なお、本プログラムでは、X,Y,Z各方向の座標値(l,m,n)を図のように設定する。) ひとつの格子点からは4方向に伝送線路が伸び、隣接する格子点とジャイレーターを介して接続される。ジャイレーターは電流と電圧を逆転する素子であって、入力側の電圧・電流を V_{in}, I_{in} 出力側を V_{out}, I_{out} とすると次式で定義される。

$$V_{out} = -I_{in} \quad I_{out} = -V_{in} \quad (\text{B-Dノード間})$$

$$V_{out} = I_{in} \quad I_{out} = V_{in} \quad (\text{他のノード間}) \quad (1)$$

このとき、各格子点における電流・電圧と電磁界成分を表2-1のように定義すると、各格子点における電流・電圧の関係式はMaxwell方程式と対応する。このように、ひとつの格子点では電界1成分・磁界2成分または電界2成分・磁界1成分が定義される。格子点から流れだす電流はその点における磁界または電界の進行波・後退波に対応し、また、格子点から伸びる線路の方向はポインティングベクトルの向きに対応する。

電磁界を求めることは、回路網の各格子点における電流・電圧を求めることであって、これはベルジェロン法を用いて計算できる。ベルジェロン法は有限長線路における伝搬特性を線路両端の電流・電圧を用いて時間軸上で定式化したもので、線路長が d 、それに対応した伝搬時間を Δt 、伝送線路の特性インピーダンスを Z_0 とすると次式で表される。

$$V(d, t + \Delta t) - Z_0 I(d, t + \Delta t) = V(0, t) - Z_0 I(0, t)$$

$$V(0, t + \Delta t) + Z_0 I(0, t + \Delta t) = V(d, t) + Z_0 I(d, t) \quad (2)$$

すなわち、回路網における Δt 時刻後の電流・電圧が隣接格子点における電流・電圧を用いて計算できることを示す。

これを概念的に示したのが図2-2である。中央にある格子点Aおよび隣接格子点をBおよびCとする。空間と時間を等差分にとれば(格子点の離散距離 Δd を観測離散時間 Δt に波が伝搬する距離と等しくとれば)時刻 t の隣接格子点の電界・磁界成分は時刻 $t + \Delta t$ には観測格子点に到達する。すなわち、点Aが示す電磁界は隣接格子点から到来する電界・磁界成分の足し合わせとして求まる。ここで隣接格子点間を伝送線路で接続した回路網を考え、各格子点における電圧と各伝送線路に流れる電流をその格子点の電界・磁界成分に対応付けたものが空間回路網法である。

この様に、この解析手法は各格子点を2次波源とみなし Δt 時間後の格子点の電磁界を計算するホイヘンスの原理の離散モデルと考えることもできる。

2.2 プログラムに用いる計算式

格子点の等価回路を図2-3に示す。格子点からは、4方向へ伝送線路が伸び、また、格子点には並列にコンデンサと抵抗が接続されている。抵抗はその格子点が代表する部分の損失に、コンデンサはその格子点における電圧が電界に対応する場合は誘電率、磁界に対応する場合は透磁率に対応する。また、励振は内部抵抗を接続した電圧源を用いる。

電流・電圧を図2-3に示した矢印の方向に定義すると、電流・電圧は時刻 Δt 前の隣接接点の値を用いて(1)(2)式より次の関係式で表される




図2-1 等価回路網

	電圧	電流		
		X方向	Y方向	Z方向
Aノード	E_y	H_z	---	$-H_x$
Dノード	E_x	---	$-H_z$	H_y
Eノード	$-E_z$	H_y	$-H_x$	---
Bノード	H_z	E_y	$-E_x$	---
Cノード	$-H_x$	---	$-E_z$	E_y
Fノード	H_y	$-E_z$	---	E_x

表2-1 電圧・電流と電界・磁界の対応表




図2-2 等価回路網




図2-3 格子点の等価回路

$$V_y(l, m, n, t) + Z_0 I_{z1}(l, m, n, t) = I_{z2}^*(l, m, n-1, t-\Delta t) + Z_0 V_x(l, m, n-1, t-\Delta t) = \Psi_1 \quad (3)$$

$$V_y(l, m, n, t) - Z_0 I_{z2}(l, m, n, t) = I_{z1}^*(l, m, n, t-\Delta t) - Z_0 V_x(l, m, n-1, t-\Delta t) = \Psi_2 \quad (4)$$

$$V_y(l, m, n, t) + Z_0 I_{x1}(l, m, n, t) = I_{x2}^*(l-1, m, n, t-\Delta t) + Z_0 V_x(l-1, m, n-1, t-\Delta t) = \Psi_3 \quad (5)$$

$$V_y(l, m, n, t) + Z_0 I_{x2}(l, m, n, t) = I_{x1}^*(l, m, n, t-\Delta t) - Z_0 V_x(l, m, n-1, t-\Delta t) = \Psi_4 \quad (6)$$

$$V_y(l, m, n, t) + Rc I_d(l, m, n, t) = V_y(l, m, n, t-\Delta t) + Rc I_d(l, m, n, t-\Delta t) = \Psi_5 \quad (7)$$

$$I_e(l, m, n, t) = G V_y(l, m, n, t) = \Psi_6 \quad (8)$$

$$I_r(l, m, n, t) = G \ln(V_y(l, m, n, t) - V_o) = \Psi_7 \quad (9)$$

ここで、 Z_0 は伝送線路の特性インピーダンスであり、自由空間の特性インピーダンスとしている。また、 V_s は励振電圧の電圧値を示す。 Rc は回路網のコンデンサの容量値であって、媒質の χ_e, χ_m を用いて次式となる。

$$Rc = \frac{\epsilon_0 \chi_e}{2\Delta d} \quad (\text{誘電率の設定}) \quad Rc = \frac{\mu_0 \chi_m}{2\Delta d} \quad (\text{透磁率の設定})$$

以上の様に、 Ψ_1 - Ψ_7 は、時刻 Δt 前の隣接接点の値を用いて計算できる。

解析空間をモデル化するときの媒質条件、境界条件として下記の6種類が考えられ、以下にその回路計算式を示す。

- 1) 空気媒質内における格子点
- 2) 誘電体および磁性体における格子点
- 3) 完全導体および磁気壁における格子点
- 4) 一端が自由空間境界となる格子点
- 5) 二端が自由空間境界となる格子点
- 6) 電圧駆動点となる格子点

(1) 空気媒質内における格子点

空気部は格子点において並列素子を持たない回路で表現される。電流の連続条件より

$$I_{z1}(l, m, n, t) - I_{z2}(l, m, n, t) + I_{x1}(l, m, n, t) - I_{x2}(l, m, n, t) = 0$$

これより電圧は次式となり、これを式(3)-(6)に代入すれば各電流値が求められる。

$$V_y(l, m, n, t) = \frac{\Psi_1 + \Psi_2 + \Psi_3 + \Psi_4}{4} \quad (10)$$

2) 誘電体および磁性体媒質内にある格子点

損失および媒質条件は、上記の式(7)(8)で示されるが、格子点が媒質境界面上にある場合は、コンデンサの容量値およびコンダクタンスは相加平均値を用いる。ここで電流の連続条件より

$$I_{z1}(l, m, n, t) - I_{z2}(l, m, n, t) + I_{x1}(l, m, n, t) - I_{x2}(l, m, n, t) + I_d(l, m, n, t) - I_c(l, m, n, t) = 0$$

よって接点における電圧は電流を消去することより

$$V_y(l, m, n, t) = \frac{Rc(\Psi_1 + \Psi_2 + \Psi_3 + \Psi_4) + Zo \Psi_5}{4Rc + Zo(1 + RcG)} \quad (11)$$

なお、損失が無い場合は次式となる。

$$V_y(l, m, n, t) = \frac{Rc(\Psi_1 + \Psi_2 + \Psi_3 + \Psi_4) + Zo \Psi_5}{4Rc + Zo} \quad (12)$$

また、誘電率や透磁率が1であって、損失のみある場合は次式となる。

$$V_y(l, m, n, t) = \frac{\Psi_1 + \Psi_2 + \Psi_3 + \Psi_4}{4 + ZoG} \quad (13)$$

これを式(3)-(8)に代入して電流値を求める。

4) 一端が自由空間境界となる格子点

自由空間は、解析領域外部から内部へエネルギーの流入がない条件として近似する。これは境界面から外に延びた線路の先端を整合負荷で終端することに対応し、格子点に並列接続抵抗を加えた等価回路となる。(図2-4) ここで電流の連続条件より

$$-I_{z2}(l, m, n, t) + I_{x1}(l, m, n, t) - I_{x2}(l, m, n, t) - I_e(l, m, n, t) = 0$$

よって接点における電圧は電流を消去することより

$$V_y(l, m, n, t) = \frac{Rc(\Psi_2 + \Psi_3 + \Psi_4) + Zo \Psi_6}{Zo + Rc(3 + ZoG)} \quad (14)$$

誘電体を含まない接点の自由空間との境界条件については次式となる。

$$V_y(l, m, n, t) = \frac{(\Psi_2 + \Psi_3 + \Psi_4)}{3 + ZoG} \quad (15)$$

これを式(3)-(8)に代入して各電流値が求められる。

7) 二端が自由空間境界となる空気媒質内における格子点

ふたつの自由空間境界面を持つ場合については境界面に延びたふたつの線路の先端を抵抗で終端することになり、並列接続抵抗として等価回路表現される。これは、先に示した式に並列コンダクタンスを付加することであり、このとき格子点における電圧は次式で得られる。

$$V_y(l, m, n, t) = \frac{(\Psi_2 + \Psi_4)}{2 + 2ZoG} \quad (16)$$

また、誘電体を含む場合については電圧は次式で得られる。

$$V_y(l, m, n, t) = \frac{Rc(\Psi_2 + \Psi_4) + Zo \Psi_6}{Zo + Rc(2 + 2ZoG)} \quad (17)$$

(9) 電圧駆動点となる格子点

励振回路は以上に示した等価回路に対して、定電圧源を接続する回路となる。この時、電源の内部抵抗は励振点における電界と磁界の比、即ちエネルギー流出量を決める重要な要素となる。この値が所望のモードに対してミスマッチ状態であるならば、励振点から観測領域へ流れ込むエネルギー量が小さくなる。電流連続の式よりノードでは次式が満たされる。

$$-I_{z2}(l,m,n,t) + I_{x1}(l,m,n,t) - I_{x2}(l,m,n,t) - I_e(l,m,n,t) = 0$$

よって、電圧は次式となる。

$$V_y(l,m,n,t) = \frac{Rc(\Psi_2 + \Psi_3 + \Psi_4) + Z_0 \Psi_6}{Z_0 + Rc(3 + Z_0 G)} \quad (18)$$

ノードの他の条件においても同様に定式化できる。

3) 完全導体および磁気壁における格子点

完全導体上においては、電界の接線成分および磁界の法線成分が0であればよい。しかしながらこの条件を表現できるのは、X-Y面ではA-D-B、Y-Z面ではA-D-B、Z-X面ではA-E-Cの面に限られる。なお、等価回路としては電気的ノードを短絡、すなわち $V_y(l,m,n,t) = 0$ と置けばよい。これにより例えばX方向から来た電流はX方向に完全に反射されることになる。




図2-4 開放領域を持つ格子点の等価回路




図2-5 金属境界を表す格子点配置

図2-5はX-Z面上に配置した金属板を表す。使用するノードはD,Eである。図の左側に示すように金属板の境界線は、その電界の接線成分が0となるノードによって規定することになる。右側はノードの選択が誤っている。これよりわかるように現時点では、斜めの導体は本解析手法では表現できない。

同様に磁気壁は、磁界の接線成分および電界の法線成分を0とするが、この条件を表現できるのは、X-Y面ではE-F-C、Y-Z面ではB-D-F、Z-X面ではA-B-Cの面に限られる。

2.3 電気的特性の計算

上記のように解析対象のモデルを作成したのち、この回路網に励振電圧を加える。電圧・電流は、励振点から広がっていき、モデル内部で反射・散乱され、解析領域の外壁で吸収されていく。

入出力端が定義されていれば、入力にパルス波を用いてインパルス応答が計算できる。このインパルス応答を用いフーリエ変換によって、周波数特性が計算できる。なお、このとき周波数特性のレファレンスが必要になるが、これはそのレファレンスモデルを作成し、前もって計算しておかねばならない。(図2-6)

また、入力に正弦波を用いれば解析領域内には電磁界の定常状態が観測できる。このときの各ノードにおける電磁界値を出力すれば、解析領域内の電磁界分布(モード)が観測できる。




図2-6 周波数特性計算

2.4 使用上の注意点及び問題点

この手法では、上記境界・媒質条件を満たすように空間のモデルをつくり、初期条件を与えて電磁界の挙動を観測する。この時格子点により空間を表すが、格子点間隔が等距離であるため解析対象物の寸法の最大公約数を格子点間隔に用いなければならない。また、斜面や曲面のような形状を直接表現できず、立方体近似しなければならない。

さらに、差分の精度を上げるためには格子点間隔を小さくしなければならない。

これらはすべて対象とする計算領域における格子点総数を多くすることになり、計算機のメモリーサイズの増大と一離散時間ごとの計算増大を招く。

この様に本解析法の利点としては、電磁界の時間的変化の観測や、広帯域な周波数特性計算ができることであるが、精度を求める場合は注意を要する。広帯域な周波数応答を大まかにつかんだり、電磁界の集中部分をつかんだりする目的に適した解析法といえよう。なお、格子点間は伝送線路で接続されているため、ローパス効果が生じることを記しておく。

本解析法では、行列計算は不要であって、加算・乗算のみで計算できるため、とくにベクトライズ機能をもつ計算機に適している。メモリー容量・計算速度というハードの制約が非常におおきい計算法である。

参考文献

- [1] 山下他：“電磁波問題の基礎解析法”、電子通信情報学界 昭62
- [2] 吉田、深井、福岡：“電磁界の接点方程式による過渡解析”、電子通信学会論文誌、vol.J63-B,no.9, Sep. 1980
- [3] 吉田、深井、福岡：“Bergeron法による2次元マクスウエル方程式の過渡解析”、電子通信学会論文誌、vol.J62-B,no.6, June 1979

3.解析プログラムの使用方法およびプログラムコードの説明

本章では、解析対象を解析領域内に一種類の誘電体材料を配置した構造物に限定した時の、プログラムコードおよび使用法を記す。本解析法での計算作業手順は以下ようになる。

- 1) モデルの作成 解析空間を格子点に置き換えたモデルを作成
- 2) 回路網の計算 モデルの一部を入力格子点として励振し、必要なデータを出力する
- 3) 計算結果の処理 計算結果を用いグラフィック表示や周波数特性を求める

1)および2)は空間回路網法計算プログラム(SNMMAIN.FOR)、3)のうちの時間応答データの周波数軸への変換は周波数変換プログラム(SNMFREQ.FOR)で行っている。これらのプログラムはVAX8650の[MUSEN2.SNM]に収納されている。

以下、3.1章に空間回路網法計算プログラム、3.2章に周波数変換プログラムのコード及び使用方法を記す。

3.1 空間回路網法計算プログラム(SNMMAIN.FOR)

本プログラムは、モデルの設定と空間回路網法計算を行う。解析対象は空間内に一種類の誘電体を配置し、透磁率はすべて1.0とした構造物である。解析領域の外壁は、X-Z平面にはD-F-E面、Y-Z平面にはA-C-E面、X-Y平面にはA-B-D面を配置し、すべて自由空間終端としている。入出力は、Z方向の両端に設定できる。本コードは、モデル作成部、計算部、出力部から成り、モデル作成部はサブルーチンとして独立している。

3.1.1 使用方法

以下の説明に従って、コードのモデル作成部および出力部を作成し、コンパイル・実行する。モデル作成部の使用例とコードは、4章に示す。

★モデル作成について

モデル作成部では、解析モデルの設定と計算条件の設定を行う。以下ではモデル作成(設定)方法について記す。

本プログラムではモデルを作成する上で、各格子点の座標値を図2-1に示した(l,m,n)の位置関係を用い、(x,y,z)=(0,0,0)に近いノードを(1,1,1)として定義している。参考として、空間における各ノードの配置番号の関係を以下に示す。

X-方向	B(x-1,y,z)--A(x,y,z)--B(x,y,z)--A(x+1,y,z) F(x-1,y,z)--E(x,y,z)--F(x,y,z)--E(x+1,y,z)
Y-方向	B(x,y-1,z)--D(x,y,z)--B(x,y,z)--D(x,y+1,z) C(x,y-1,z)--E(x,y,z)--C(x,y,z)--E(x,y+1,z)
Z-方向	C(x,y,z-1)--A(x,y,z)--C(x,y,z)--A(x,y,z+1) F(x,y,z-1)--D(x,y,z)--F(x,y,z)--D(x,y,z+1)

モデルを表現するため、各格子点の誘電率を設定する配列JAA, JAD, JAEおよび金属を設定する配列AAK, ADK, AEKを割当て、任意の解析モデルに対処出来るようにしている。これらは、各座標に対応するA,D,Eノードの条件を決める。(解析対象においては磁気条件の設定が不要なためB,C,Fノードに対応する配列はない。)

誘電体設定の配列には、次の数値を入力する。

- 1: 空気
- 2: 空気・誘電体境界面
- 3: 誘電体

金属設定の配列には、次の数値を入力する。

- 0: 金属
- 1: 金属なし

なお、金属は下記の面にあるノードにおいて設定する。

- X-Z平面はD-F-E面。金属の境界線はX方向はEノード、Z方向はDノードとなる。
- Y-Z平面はA-C-E面。金属の境界線はY方向はEノード、Z方向はAノードとなる。
- X-Y平面はA-B-D面。金属の境界線はX方向はAノード、Y方向はDノードとなる。

本プログラムにおいて励振点はZ=1のX-Y面にあるA,Dノードに対する電界励振としている。励振点は配列IAA, IADを用いて設定する。なお、励振には、レイズドコサイン波を用いている。(正弦波を用いる場合は、計算部の修正が必要であるが、この配列はそのまま使える。)

- 0: 励振なし(自由空間条件)
- 1: 正方向電界励振
- 1: 負方向電界励振

その他、変数として以下の物を用いており、解析モデルに合わせて設定する。

- | | | |
|-------|-------------|-------------|
| EPA | : 誘電率 | 誘電体の誘電率 |
| VO | : 電源電圧 | 励振電源の電圧値 |
| ZIN | : 入力インピーダンス | 励振電源の内部抵抗値 |
| INTIM | : 入力波のパルス幅 | レイズドコサインの周期 |
| ICALT | : 計算回数 | 計算(観測)回数の設定 |

なお、メインプログラムのPARAMETER文にて、全体の解析領域サイズIXM, IYM, IZM(各方向の計算領域数)を設定している。

以上、モデルの設定について述べた。サブルーチン部のソースコードは、4章に示している。なお、解析領域(配列)が大きくなる場合はサブルーチン部を独立させたほうが、計算機に対する負荷が小さく、好ましい。

★出力部の作成

出力部は計算ルーチンの最後に設定している。出力データは各時刻ごとの電磁界値をファイルなどに出力することになる。従って、Sパラメータが知りたいならば、注目する1点の電界値を1計算回ごとに出力することになるし、電磁界の挙動が知りたい場合は注目する範囲の電界値を出力することになる。

計算ルーチン内では、電磁界値は下に示すような配列位置をとっている。これを参照しながら、電磁界値やその他必要なデータを出力するように出力部の修正を行う。

ノード記号	電圧値	電流値				誘電体		
		-X方向	+X方向	-Y方向	+Y方向		-Z方向	+Z方向
Aノード AA	(1,*) Ey	(4,*) Hx	(5,*) Hx	----- -----	----- -----	(2,*) Hz	(3,*) Hz	(6,*)
Dノード AD	(1,*) Ex	----- -----	----- -----	(4,*) Hy	(5,*) Hy	(2,*) Hz	(3,*) Hz	(6,*)
Eノード AE	(1,*) Ez	(2,*) Hx	(3,*) Hx	(4,*) Hy	(5,*) Hy	----- -----	----- -----	(6,*)
Bノード AB	(1,*) Hz	(2,*) Ex	(3,*) Ex	(4,*) Ey	(5,*) Ey	----- -----	----- -----	-----
Cノード AC	(1,*) Hx	----- -----	----- -----	(4,*) Ey	(5,*) Ey	(2,*) Ez	(3,*) Ez	-----
Fノード AF	(1,*) Hy	(4,*) Ex	(5,*) Ex	----- -----	----- -----	(2,*) Ez	(3,*) Ez	-----

3.1.2 プログラムコード

以下では、プログラムコードの内容について簡単に記すとともに、コードリストを示す。

本プログラムでは、上記モデルを用いて2章に示した式に従い計算を実行している。なお、自由空間の誘電率・透磁率は計算時間の短縮を行うため規格化している。 $(\epsilon_0 = \mu_0 = 1, Z_0 = 1)$ 計算はまず、時間 Δt 前の値を用いて2章に示した Ψ を計算し、配列 AAT, ABT, ACT, ADT, AET, AFT に格納する。次ぎにこの値を用いて、実際の時間の電圧・電流を計算し、配列 AA, AB, AC, AD, AE, AF に格納している。なお、以下のコードでは、計算機のベクトル計算機能を発揮させるため、解析領域の外部にダミーノードを配置している。

本解析法において電磁界値、すなわち各格子点の電流・電圧値は、その隣接格子点の電流・電圧値を用いた加減乗算によって、1時間ステップごとの結果が得られる。従って、解析対象物に応じたメモリー容量の設定や演算回数低減となる計算ルーチンの組替えを行ったほうがメモリー容量の節約および計算時間の短縮の点から有利である。コメント文にある指示に従いチューニングをするだけで計算時間が1/20以下になる。

例えば、各計算時間ごとの各ノード電流に対応する電界・磁界値が不要な場合は、2章に示した Ψ を計算するだけでよい。よって、各時間ステップごとに、電圧・電流を求めず、いきなり Ψ 値のみを計算することにより、計算量とメモリーの大幅な節約ができる。

これらの点を考えながら、最適なコードへモディファイしてほしい。

```
C*****
C**** 空間回路網法による計算
C****
C****                                     作成 1987年12月 4日 角田紀久夫
C****                                     修正 1988年 5月 5日 角田紀久夫
C****
C****   このプログラムは、空間回路網法による電磁界解析プログラムである。
C****   プログラムは汎用形であり、計算時にはプログラム効率をあげるため
C****   次の点をモディファイし、チューニングをすれば計算時間が1/20以下
C****   になる。
C****   0. データ作成部分を別プログラムに分離し、プログラムサイズを
C****       縮小すること。
C****   1. CPU能力を考慮し、変数領域の縮小をはかる。
C****   2. 2次元以上の配列は1次元化する。場合によっては配列を複数に
C****       分割する。
C****   3. DO文の中に含まれているIF文をループ外に出す。
C****   4. 計算式における演算回数を少なくなるように考慮する。
C****   5. 定数への置き換えの可否を検討する。
C****
C****
C*****
C
C ○自由空間の誘電率・透磁率は計算時間の短縮を行うため規格化している。
C   誘電率,透磁率,光速は 1.0
C ○誘電体基板は1種類とする。また、このプログラムでは透磁率は設定できない。
C ○金属壁,入出力面,自由空間終端等境界条件を設定する面は次の配置となる。
C   X-Z 平面には D-F-E 面
C   Y-Z 平面には A-C-E 面
C   X-Y 平面には A-B-D 面
C ○各ノードにおける電気壁境界条件は配列 AAK(*,*,*),ADK(*,*,*),AEK(*,*,*)
C   に設定される。なおこのメモリーは計算時には、特定の値に再設定される。
C   1 : 金属      2 : 拘束条件なし
C ○各ノードにおける誘電率条件は配列 JAA(*,*,*),JAD(*,*,*),JAE(*,*,*)
```

C に設定される。

C 1 : 自由空間 2 : 自由空間-誘電体境界面 3 : 誘電体

C ○入力は Z=0 の X-Y 面にある A ノード, D ノードの電界励振とし, 励振条件は
C 配列 IAA(*,*), IAD(*,*) に設定される。

C 0 : 励振なし 1 : 正方向電圧励振 -1 : 負方向電圧励振

C ○Z = NV の X-Y 面にある A ノード, D ノードの終端抵抗は配列 LOADA(*,*),
C LOADD(*,*) に設定される。

C 0 : 自由空間抵抗値 1 : 指定負荷抵抗値

C ○出力ノードは任意の点をアウトプット用ルーチンにて随時設定する。

C ○空間の分割数は上記面の繰り返し数として定義する。

C ノード間の間隔を Δd とし, 解析対象空間のサイズを X, Y, Z 各方向
C $2 \times \Delta d \times LV, 2 \times \Delta d \times MV, 2 \times \Delta d \times NV$ とすると, 各ノードの配列数を
C 次の様に設定する必要がある。

C A ノード (LV, MV-1, NV) : D ノード (LV-1, MV, NV) : E ノード (LV, MV, NV-1)
C F ノード (LV-1, MV, NV-1) : B ノード (LV-1, MV-1, NV) : C ノード (LV, MV-1, NV-1)

C ○空間における各ノードの配置番号は次の位置関係になる。

C X-方向 B(x-1, y, z) -- A(x, y, z) -- B(x, y, z) -- A(x+1, y, z)
C F(x-1, y, z) -- E(x, y, z) -- F(x, y, z) -- E(x+1, y, z)

C Y-方向 B(x, y-1, z) -- D(x, y, z) -- B(x, y, z) -- D(x, y+1, z)
C C(x, y-1, z) -- E(x, y, z) -- C(x, y, z) -- E(x, y+1, z)

C Z-方向 C(x, y, z-1) -- A(x, y, z) -- C(x, y, z) -- A(x, y, z+1)
C F(x, y, z-1) -- D(x, y, z) -- F(x, y, z) -- D(x, y, z+1)

C ○各ノードにおける電圧・電流と電界・磁界および配列との関係はつぎの用になる。

ノード記号	電圧値	電流値						誘電体
		-X方向	+X方向	-Y方向	+Y方向	-Z方向	+Z方向	
Aノード AA	(1,*)	(4,*)	(5,*)	-----	-----	(2,*)	(3,*)	(6,*)
	Ey	Hx		-----	-----	Hx		Hx
Dノード AD	(1,*)	-----	-----	(4,*)	(5,*)	(2,*)	(3,*)	(6,*)
	Ex	-----	-----	Hy	-----	Hz	-----	Hz
Eノード AE	(1,*)	(2,*)	(3,*)	(4,*)	(5,*)	-----	-----	(6,*)
	Ez	Hx	Hy	-----	-----	-----	-----	-----
Bノード AB	(1,*)	(2,*)	(3,*)	(4,*)	(5,*)	-----	-----	-----
	Hx	Ex	Ey	-----	-----	-----	-----	-----
Cノード AC	(1,*)	-----	-----	(4,*)	(5,*)	(2,*)	(3,*)	-----
	Hx	-----	-----	Ey	-----	Ez	-----	-----
Fノード AF	(1,*)	(4,*)	(5,*)	-----	-----	(2,*)	(3,*)	-----
	Hy	Ex	-----	-----	-----	Ez	-----	-----

C *****

C 変数リスト

C IXM, IYM, IZM : 配列(計算)領域のディメンジョン

C LV, MV, NV : 計算領域の最大値

C J, L, M, N : 配列計算時のカウンター (未知数番号, X, Y, Z)

C AAT(6,*,*,*), ADT(6,*,*,*), AET(6,*,*,*) : Δt 時間前の電気ノードの計算値

C ABT(5,*,*,*), ACT(5,*,*,*), AFT(5,*,*,*) : Δt 時間前の磁気ノードの計算値

C IAA(*,*), IAD(*,*) : 電源接続条件

C JAA(*,*), JAD(*,*), JAE(*,*) : 電気ノードの誘電体条件

C AA(6,*,*,*), AD(6,*,*,*), AE(6,*,*,*) : 現在時刻の電気ノードの値

C AB(5,*,*,*), AC(5,*,*,*), AF(5,*,*,*) : 現在時刻の磁気ノードの値

C AAK(*,*), ADK(*,*), AEK(*,*) : 電気ノードの境界条件

C AAKK(*,*), ADKK(*,*) : 入力面の電気ノードの境界条件

C SHUNT(2,3) : 電気ノード計算時の境界条件による係数値

C SHUNT(1,*) == 金属境界面

C SHUNT(1.1) : 空気内部にあるノードの電圧を求める時の係数

C SHUNT(1.2) : 誘電体-空気境界面にあるノードの電圧を求める時の係数

C SHUNT(1.3) : 誘電体内部にあるノードの電圧を求める時の係数

C SHUNT(2,*) == 4端子使用回路

```

C SHUNT(2,1) : 空気内部にあるノードの電圧を求める時の係数
C SHUNT(2,2) : 誘電体-空気境界面にあるノードの電圧を求める時の係数
C SHUNT(2,3) : 誘電体内部にあるノードの電圧を求める時の係数
C RC : 誘電体内部にある電氣的ノードにて誘電率をしめす回路定数
C RCC : 誘電体-空気境界面にある電氣的ノードにて誘電率をしめす回路定数
C EPA : 誘電率
C PAI : 円周率
C IT : 計算回数のカウンター
C GRR(3) : 終端コンダクタンス値
C          1.自由空間内 2.誘電体-空気境界 3.誘電体内
C ICHEK : 計算時の数値メモリー
C BETA : 電源の1ステップ時間あたりの位相変化量
C VO : 電源電圧
C VO : 電源最大振幅
C JJOK : グラフ表示間隔のカウンター
C INTIM ; 励振波周期
C ICALT : 計算回数
C
C*****
C
C          IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C ***** メモリーのサイズ指定
C          計算する領域のサイズを入力する。
C          PARAMETER (IXM=41,IYM=11,IZM=81)
C ***** 配列宣言
C          DIMENSION
C          # AA(6,IXM,IYM,IZM), AD(6,IXM,IYM,IZM), AE(6,IXM,IYM,IZM),
C          # AB(5,0:IXM,0:IYM,0:IZM),AC(5,0:IXM,0:IYM,0:IZM),
C          # AF(5,0:IXM,0:IYM,0:IZM),
C          # AAT(6,IXM,IYM,IZM), ADT(6,IXM,IYM,IZM),
C          # AET(6,IXM,IYM,IZM), ABT(5,IXM,IYM,IZM),
C          # ACT(5,IXM,IYM,IZM),AFT(5,IXM,IYM,IZM),
C          # AAK(IXM,IYM,IZM),ADK(IXM,IYM,IZM),AEK(IXM,IYM,IZM),
C          # JAA(IXM,IYM,IZM),JAD(IXM,IYM,IZM),JAE(IXM,IYM,IZM),
C          # VAA(IXM,IYM),VAD(IXM,IYM),LOADA(IXM,IYM),LOADD(IXM,IYM),
C          # AAKK(IXM,IYM),ADKK(IXM,IYM),
C          # SHNT4(3),SHNT3(3),SHNT2(3),SHNT1(3),GRR(3),VSHNT(3)
C
C          LV=IXM
C          MV=IYM
C          NV=IZM
C ***** 全ノードの値の初期設定
C          DO N=1,NV
C            DO M=1,MV
C              DO L=1,LV
C                DO J=1,6
C                  AA(J,L,M,N)= 0.0
C                  AD(J,L,M,N)= 0.0
C                  AE(J,L,M,N)= 0.0
C                ENDDO
C              DO J=1,5
C                AB(J,L,M,N)= 0.0
C                AC(J,L,M,N)= 0.0
C                AF(J,L,M,N)= 0.0
C              ENDDO
C            ENDDO
C          ENDDO
C        ENDDO

```

```

C
C ***** データの読み込み
      CALL SEIBI(IXM,IYM,IZM,EPA,VO,ZIN,INTIM,ICALT,IDTIM,
      # AAK,ADK,AEK,VAA,VAD,LOADA,LOADD,JAA,JAD,JAЕ,MDV)
C
      MDAT1=MDV-1
      LDAT1=15
      LDAT2=21
      LDAT3=27
      INCHR=400
C ***** 誘電率表示用容量値・電源の内部コンダクタンスの計算
C   RC : 誘電体内部にある電気的ノードにて誘電率をしめす回路定数
C   RCC: 誘電体・空気境界面にある電気的ノードにて誘電率をしめす回路定数
C   ZA : 自由空間における特性アドミタンス (=1.0)
C   ZAD: 誘電体・空気境界面における特性アドミタンス
C   ZD : 誘電体内部における特性アドミタンス
      RC=0.125/(EPA-1.0)
      RCC=RC*2.0
      GRR(1)=1.0
      GRR(2)=SQRT(0.5*(EPA+1.0))
      GRR(3)=SQRT(EPA)
      ZIN=377.0/ZIN
C
C ***** 電気的ノードの基本集中回路定数素子値の計算
C   電圧計算時の分母の値
C
C   SHUNT(*) === 4端子使用回路
C   SHUNT(1):空気   SHUNT(2):誘電体・空気境界面   SHUNT(3):誘電体内部
      SHNT4(1)=0.25
      SHNT4(2)=1.0/(1.0+4.0*RCC)
      SHNT4(3)=1.0/(1.0+4.0*RC)
C
C   SHNT3(*) === 3端子使用回路 -- 1端子終端
C   SHUNT(2,1):空気   SHUNT(2,2):誘電体・空気境界面   SHUNT(2,3):誘電体内部
      SHNT3(1)=1.0/(3.0+GRR(1))
      SHNT3(2)=1.0/(1.0+3.0*RCC+GRR(2)*RCC)
      SHNT3(3)=1.0/(1.0+3.0*RC+GRR(3)*RC)
C
C   SHNT2(*) === 2端子使用回路 -- 2端子終端
C   SHUNT(3,1):空気   SHUNT(3,2):誘電体・空気境界面   SHUNT(3,3):誘電体内部
      SHNT2(1)=1.0/(2.0+2.0*GRR(1))
      SHNT2(2)=1.0/(1.0+2.0*RCC+2.0*GRR(2)*RCC)
      SHNT2(3)=1.0/(1.0+2.0*RC+2.0*GRR(3)*RC)
C
C   SHNT1(*) === 3端子使用回路 -- 1端子終端,入力面
C   SHUNT(2,1):空気   SHUNT(2,2):誘電体・空気境界面   SHUNT(2,3):誘電体内部
      SHNT1(1)=1.0/(3.0+ZIN)
      SHNT1(2)=1.0/(1.0+3.0*RCC+ZIN*RCC)
      SHNT1(3)=1.0/(1.0+3.0*RC+ZIN*RC)
C
C ***** 各電気的ノードの回路定数登録
C   Aノードは Z=1,Z=NV および X=1,X=LV にて自由終端
      N=1
      DO M=1,MV-1
        L=1
        AAK(L,M,N)=AAK(L,M,N)*SHNT2(JAA(L,M,N))
        DO L=2,LV-1
          AAK(L,M,N)=AAK(L,M,N)*SHNT3(JAA(L,M,N))

```

```

        ENDDO
        L=LV
        AAK(L,M,N)=AAK(L,M,N)*SHNT2(JAA(L,M,N))
        ENDDO
C
DO N=2,NV-1
DO M=1,MV-1
L=1
AAK(L,M,N)=AAK(L,M,N)*SHNT3(JAA(L,M,N))
DO L=2,LV-1
AAK(L,M,N)=AAK(L,M,N)*SHNT4(JAA(L,M,N))
ENDDO
L=LV
AAK(L,M,N)=AAK(L,M,N)*SHNT3(JAA(L,M,N))
ENDDO
ENDDO
C
N=NV
DO M=1,MV-1
L=1
AAK(L,M,N)=AAK(L,M,N)*SHNT2(JAA(L,M,N))
DO L=2,LV-1
AAK(L,M,N)=AAK(L,M,N)*SHNT3(JAA(L,M,N))
ENDDO
L=LV
AAK(L,M,N)=AAK(L,M,N)*SHNT2(JAA(L,M,N))
ENDDO
C
Dノ下は Z=1,Z=NV および Y=1,Y=MVにて自由終端
N=1
M=1
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT2(JAD(L,M,N))
ENDDO
DO M=2,MV-1
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT3(JAD(L,M,N))
ENDDO
ENDDO
M=MV
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT2(JAD(L,M,N))
ENDDO
C
DO N=2,NV-1
M=1
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT3(JAD(L,M,N))
ENDDO
DO M=2,MV-1
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT4(JAD(L,M,N))
ENDDO
ENDDO
M=MV
DO L=1,LV-1
ADK(L,M,N)=ADK(L,M,N)*SHNT3(JAD(L,M,N))
ENDDO

```

```

        ENDDO
C
N=NV
M=1
DO L=1,LV-1
  ADK(L,M,N)=ADK(L,M,N)*SHNT2(JAD(L,M,N))
  ENDDO
DO M=2,MV-1
  DO L=1,LV-1
    ADK(L,M,N)=ADK(L,M,N)*SHNT3(JAD(L,M,N))
    ENDDO
  ENDDO
M=MV
DO L=1,LV-1
  ADK(L,M,N)=ADK(L,M,N)*SHNT2(JAD(L,M,N))
  ENDDO
C
C Eノドは X=1,X=LV および Y=1,Y=MV にて自由終端
DO N=1,NV-1
  M=1
  L=1
  AEK(L,M,N)=AEK(L,M,N)*SHNT2(JAE(L,M,N))
  DO L=2,LV-1
    AEK(L,M,N)=AEK(L,M,N)*SHNT3(JAE(L,M,N))
    ENDDO
  L=LV
  AEK(L,M,N)=AEK(L,M,N)*SHNT2(JAE(L,M,N))
  DO M=2,MV-1
    L=1
    AEK(L,M,N)=AEK(L,M,N)*SHNT3(JAE(L,M,N))
    DO L=2,LV-1
      AEK(L,M,N)=AEK(L,M,N)*SHNT4(JAE(L,M,N))
      ENDDO
    L=LV
    AEK(L,M,N)=AEK(L,M,N)*SHNT3(JAE(L,M,N))
    ENDDO
C
M=MV
L=1
AEK(L,M,N)=AEK(L,M,N)*SHNT2(JAE(L,M,N))
DO L=2,LV-1
  AEK(L,M,N)=AEK(L,M,N)*SHNT3(JAE(L,M,N))
  ENDDO
L=LV
AEK(L,M,N)=AEK(L,M,N)*SHNT2(JAE(L,M,N))
ENDDO
C ***** 終端
DO M=1,MV-1
  DO L=1,LV
    IF(LOADA(L,M).EQ.1)AAK(L,M,NV)=SHNT1(JAA(L,M,NV))
    ENDDO
  ENDDO
C
DO M=1,MV
  DO L=1,LV-1
    IF(LOADD(L,M).EQ.1)ADK(L,M,NV)=SHNT1(JAD(L,M,NV))
    ENDDO
  ENDDO

```

```

C ***** 電源
DO M=1,MV-1
DO L=1,LV
AAKK(L,M)=AAK(L,M,1)
ENDDO
ENDDO
DO M=1,MV-1
DO L=2,LV-1
IF(VAA(L,M).EQ.1)AAKK(L,M)=SHNT1(JAA(L,M,1))
ENDDO
ENDDO

C
DO M=1,MV
DO L=1,LV-1
ADKK(L,M)=ADK(L,M,1)
ENDDO
ENDDO
DO M=2,MV-1
DO L=1,LV-1
IF(VAD(L,M).EQ.1)ADKK(L,M)=SHNT1(JAD(L,M,1))
ENDDO
ENDDO

C
VSHNT(1)=ZIN
VSHNT(2)=ZIN*RCC
VSHNT(3)=ZIN*RC
DO M=1,MV-1
DO L=1,LV
VAA(L,M)=VAA(L,M)*VSHNT(JAA(L,M,1))
ENDDO
ENDDO
DO M=1,MV
DO L=1,LV-1
VAD(L,M)=VAD(L,M)*VSHNT(JAD(L,M,1))
ENDDO
ENDDO

C ***** 電源電圧設定
PAI=3.1415
BETA=2.0*PAI/INTIM
C ===== 計算ルーチン
IT=-1
10 CONTINUE
IF(IT.GE.ICALT+1)GOTO 20
IT=IT+1
V0=0.0
IF(IT.LT.INTIM) V0=V0*(1.0-COS(BETA*IT))

C
C ***** 各ノードの前計算 *****
C *** B :磁気ノード (1,1,1) から(LV-1,MV-1,NV)
C 全範囲で通常計算
C 全空間
DO N=1,NV
DO M=1,MV-1
DO L=1,LV-1
ABT(1,L,M,N)= AA(5,L,M,N) +AA(1,L,M,N)
ABT(2,L,M,N)= AA(4,L+1,M,N)-AA(1,L+1,M,N)
ABT(3,L,M,N)=-AD(5,L,M,N) -AD(1,L,M,N)
ABT(4,L,M,N)=-AD(4,L,M+1,N)+AD(1,L,M+1,N)

```

```

ABT(5,L,M,N)=ABT(1,L,M,N)+ABT(2,L,M,N)+ABT(3,L,M,N)+ABT(4,L,M,N)
  ENDDO
  ENDDO
  ENDDO
C
C
C C :磁気ノード (1,1,1) から(LV,MV-1,NV-1)
C   全範囲で通常計算
C 全空間
  DO N=1,NV-1
  DO M=1,MV-1
  DO L=1,LV
  ACT(1,L,M,N)= AA(3,L,M,N) +AA(1,L,M,N)
  ACT(2,L,M,N)= AA(2,L,M,N+1)-AA(1,L,M,N+1)
  ACT(3,L,M,N)= AE(5,L,M,N) +AE(1,L,M,N)
  ACT(4,L,M,N)= AE(4,L,M+1,N)-AE(1,L,M+1,N)
  ACT(5,L,M,N)=ACT(1,L,M,N)+ACT(2,L,M,N)+ACT(3,L,M,N)+ACT(4,L,M,N)
  ENDDO
  ENDDO
  ENDDO
C
C
C F :磁気ノード (1,1,1) から(LV-1,MV,NV-1)
C   全範囲で通常計算
C 全空間
  DO N=1,NV-1
  DO M=1,MV
  DO L=1,LV-1
  AFT(1,L,M,N)= AD(3,L,M,N) +AD(1,L,M,N)
  AFT(2,L,M,N)= AD(2,L,M,N+1)-AD(1,L,M,N+1)
  AFT(3,L,M,N)= AE(3,L,M,N) +AE(1,L,M,N)
  AFT(4,L,M,N)= AE(2,L+1,M,N)-AE(1,L+1,M,N)
  AFT(5,L,M,N)=AFT(1,L,M,N)+AFT(2,L,M,N)+AFT(3,L,M,N)+AFT(4,L,M,N)
  ENDDO
  ENDDO
  ENDDO
C
C
C
C A :電気ノード (1,1,1) から(LV,MV-1,NV)
  DO N=1,NV
  DO M=1,MV-1
  DO L=1,LV
  AAT(1,L,M,N)= AC(3,L,M,N-1)+AC(1,L,M,N-1)
  AAT(2,L,M,N)= AC(2,L,M,N) -AC(1,L,M,N)
  AAT(3,L,M,N)= AB(3,L-1,M,N)+AB(1,L-1,M,N)
  AAT(4,L,M,N)= AB(2,L,M,N) -AB(1,L,M,N)
  AAT(6,L,M,N)=AAT(1,L,M,N)+AAT(2,L,M,N)+AAT(3,L,M,N)+AAT(4,L,M,N)
  IF(JAA(L,M,N).NE.1) AAT(5,L,M,N)= AA(1,L,M,N) +AA(6,L,M,N)
  IF(JAA(L,M,N).EQ.2) AAT(6,L,M,N)=RCC*AAT(6,L,M,N)+AAT(5,L,M,N)
  IF(JAA(L,M,N).EQ.3) AAT(6,L,M,N)=RC*AAT(6,L,M,N)+AAT(5,L,M,N)
  ENDDO
  ENDDO
  ENDDO
C
C
C D :電気ノード (1,1,1) から(LV-1,MV,NV)
C

```

```

DO N=1,NV
DO M=1,MV
DO L=1,LV-1
ADT(1,L,M,N)= AF(3,L,M,N-1)+AF(1,L,M,N-1)
ADT(2,L,M,N)= AF(2,L,M,N) -AF(1,L,M,N)
ADT(3,L,M,N)=-AB(5,L,M-1,N)-AB(1,L,M-1,N)
ADT(4,L,M,N)=-AB(4,L,M,N) +AB(1,L,M,N)
ADT(6,L,M,N)=ADT(1,L,M,N)+ADT(2,L,M,N)+ADT(3,L,M,N)+ADT(4,L,M,N)
IF(JAD(L,M,N).NE.1) ADT(5,L,M,N)= AD(1,L,M,N)+AD(6,L,M,N)
IF(JAD(L,M,N).EQ.2) ADT(6,L,M,N)=RCC*ADT(6,L,M,N)+ADT(5,L,M,N)
IF(JAD(L,M,N).EQ.3) ADT(6,L,M,N)=RC*ADT(6,L,M,N)+ADT(5,L,M,N)
ENDDO
ENDDO
ENDDO

```

```

C
C E :電気ノード (1,1,1) から(LV,MV,NV-1)
C

```

```

DO N=1,NV-1
DO M=1,MV
DO L=1,LV
AET(1,L,M,N)= AF(5,L-1,M,N)+AF(1,L-1,M,N)
AET(2,L,M,N)= AF(4,L,M,N) -AF(1,L,M,N)
AET(3,L,M,N)= AC(5,L,M-1,N)+AC(1,L,M-1,N)
AET(4,L,M,N)= AC(4,L,M,N) -AC(1,L,M,N)
AET(6,L,M,N)=AET(1,L,M,N)+AET(2,L,M,N)+AET(3,L,M,N)+AET(4,L,M,N)
IF(JAE(L,M,N).NE.1) AET(5,L,M,N)=AE(1,L,M,N)+AE(6,L,M,N)
IF(JAE(L,M,N).EQ.2) AET(6,L,M,N)=RCC*AET(6,L,M,N)+AET(5,L,M,N)
IF(JAE(L,M,N).EQ.3) AET(6,L,M,N)=RC*AET(6,L,M,N)+AET(5,L,M,N)
ENDDO
ENDDO
ENDDO

```

```

C
C
C ***** 実時間計算 *****
C

```

```

C *** B :磁気ノード (1,1,1) から(LV-1,MV-1,NV)
C 全範囲で通常計算
C 全体

```

```

DO N=1,NV
DO M=1,MV-1
DO L=1,LV-1
AB(1,L,M,N)= ABT(5,L,M,N)*0.25
AB(2,L,M,N)= ABT(1,L,M,N)-AB(1,L,M,N)
AB(3,L,M,N)=-ABT(2,L,M,N)+AB(1,L,M,N)
AB(4,L,M,N)= ABT(3,L,M,N)-AB(1,L,M,N)
AB(5,L,M,N)=-ABT(4,L,M,N)+AB(1,L,M,N)
ENDDO
ENDDO
ENDDO

```

```

C
C
C C :磁気ノード (1,1,1) から(LV,MV-1,NV-1)
C 全範囲で通常計算
C 全体

```

```

DO N=1,NV-1
DO M=1,MV-1
DO L=1,LV
AC(1,L,M,N)= ACT(5,L,M,N)*0.25

```

```

AC(2,L,M,N)= ACT(1,L,M,N)-AC(1,L,M,N)
AC(3,L,M,N)=-ACT(2,L,M,N)+AC(1,L,M,N)
AC(4,L,M,N)= ACT(3,L,M,N)-AC(1,L,M,N)
AC(5,L,M,N)=-ACT(4,L,M,N)+AC(1,L,M,N)
  ENDDO
  ENDDO
ENDDO
C
C
C F :磁気ノード (1,1,1) から(LV-1,MV,NV-1)
CC 全体
DO N=1,NV-1
DO M=1,MV
DO L=1,LV-1
  AF(1,L,M,N)= AFT(5,L,M,N)*0.25
  AF(2,L,M,N)= AFT(1,L,M,N)-AF(1,L,M,N)
  AF(3,L,M,N)=-AFT(2,L,M,N)+AF(1,L,M,N)
  AF(4,L,M,N)= AFT(3,L,M,N)-AF(1,L,M,N)
  AF(5,L,M,N)=-AFT(4,L,M,N)+AF(1,L,M,N)
  ENDDO
  ENDDO
ENDDO
C
C
C A :電気ノード (1,1,1) から(LV,MV-1,NV)
C
  N=1
  IF(IT.LE.INCHR)THEN
    DO M=1,MV-1
    DO L=1,LV
      AA(1,L,M,N)=(AAT(6,L,M,N)+VO*VAA(L,M))*AAKK(L,M)
      AA(2,L,M,N)=(AAT(1,L,M,N)-AA(1,L,M,N))
      AA(3,L,M,N)=(-AAT(2,L,M,N)+AA(1,L,M,N))
      AA(4,L,M,N)=(AAT(3,L,M,N)-AA(1,L,M,N))
      AA(5,L,M,N)=(-AAT(4,L,M,N)+AA(1,L,M,N))
      IF(JAA(L,M,N).NE.1) AA(6,L,M,N)= AA(1,L,M,N)-AAT(5,L,M,N)
    ENDDO
  ENDDO
  ELSE
    DO M=1,MV-1
    DO L=1,LV
      AA(1,L,M,N)=AAT(6,L,M,N)*AAK(L,M,N)
      AA(2,L,M,N)=(AAT(1,L,M,N)-AA(1,L,M,N))
      AA(3,L,M,N)=(-AAT(2,L,M,N)+AA(1,L,M,N))
      AA(4,L,M,N)=(AAT(3,L,M,N)-AA(1,L,M,N))
      AA(5,L,M,N)=(-AAT(4,L,M,N)+AA(1,L,M,N))
      IF(JAA(L,M,N).NE.1) AA(6,L,M,N)= AA(1,L,M,N)-AAT(5,L,M,N)
    ENDDO
  ENDDO
  ENDDIF
C
  DO N=2,NV
  DO M=1,MV-1
  DO L=1,LV
    AA(1,L,M,N)=AAT(6,L,M,N)*AAK(L,M,N)
    AA(2,L,M,N)=(AAT(1,L,M,N)-AA(1,L,M,N))
    AA(3,L,M,N)=(-AAT(2,L,M,N)+AA(1,L,M,N))
    AA(4,L,M,N)=(AAT(3,L,M,N)-AA(1,L,M,N))

```

```

AA(5,L,M,N)=(-AAT(4,L,M,N)+AA(1,L,M,N))
IF(JAA(L,M,N).NE.1) AA(6,L,M,N)= AA(1,L,M,N)-AAT(5,L,M,N)
  ENDDO
  ENDDO
  ENDDO
C
C
C
C D :電気ノード (1,1,1) から(LV-1,MV,NV)
C D NODE
  N=1
  IF(IT.LE.INCHR)THEN
    DO M=1,MV
      DO L=1,LV-1
        AD(1,L,M,N)=(ADT(6,L,M,N)+V0*VAD(L,M))*ADKK(L,M)
        AD(2,L,M,N)=(ADT(1,L,M,N)-AD(1,L,M,N))
        AD(3,L,M,N)=(-ADT(2,L,M,N)+AD(1,L,M,N))
        AD(4,L,M,N)=(ADT(3,L,M,N)-AD(1,L,M,N))
        AD(5,L,M,N)=(-ADT(4,L,M,N)+AD(1,L,M,N))
        IF(JAD(L,M,N).NE.1) AD(6,L,M,N)= AD(1,L,M,N)-ADT(5,L,M,N)
          ENDDO
          ENDDO
          ELSE
            DO M=1,MV
              DO L=1,LV-1
                AD(1,L,M,N)=ADT(6,L,M,N)*ADK(L,M,N)
                AD(2,L,M,N)=(ADT(1,L,M,N)-AD(1,L,M,N))
                AD(3,L,M,N)=(-ADT(2,L,M,N)+AD(1,L,M,N))
                AD(4,L,M,N)=(ADT(3,L,M,N)-AD(1,L,M,N))
                AD(5,L,M,N)=(-ADT(4,L,M,N)+AD(1,L,M,N))
                IF(JAD(L,M,N).NE.1) AD(6,L,M,N)= AD(1,L,M,N)-ADT(5,L,M,N)
                  ENDDO
                  ENDDO
                  ENDDO
C
DO N=2,NV
  DO M=1,MV
    DO L=1,LV-1
      AD(1,L,M,N)=ADT(6,L,M,N)*ADK(L,M,N)
      AD(2,L,M,N)=(ADT(1,L,M,N)-AD(1,L,M,N))
      AD(3,L,M,N)=(-ADT(2,L,M,N)+AD(1,L,M,N))
      AD(4,L,M,N)=(ADT(3,L,M,N)-AD(1,L,M,N))
      AD(5,L,M,N)=(-ADT(4,L,M,N)+AD(1,L,M,N))
      IF(JAD(L,M,N).NE.1) AD(6,L,M,N)= AD(1,L,M,N)-ADT(5,L,M,N)
        ENDDO
        ENDDO
        ENDDO
C
C
C E :電気ノード (1,1,1) から(LV,MV,NV-1)
C
DO N=1,NV-1
  DO M=1,MV
    DO L=1,LV
      AE(1,L,M,N)=AET(6,L,M,N)*AEK(L,M,N)
      AE(2,L,M,N)=(AET(1,L,M,N)-AE(1,L,M,N))
      AE(3,L,M,N)=(-AET(2,L,M,N)+AE(1,L,M,N))
      AE(4,L,M,N)=(AET(3,L,M,N)-AE(1,L,M,N))

```

```
AE(5,L,M,N)=(-AET(4,L,M,N)+AE(1,L,M,N))
IF(JAE(L,M,N).NE.1) AE(6,L,M,N)= AE(1,L,M,N)-AET(5,L,M,N)
ENDDO
ENDDO
ENDDO
```

C

C *****出力部

C ***** テータ アウトスト

C ***** 電界強度データ出力用

```
WRITE(60,*)IT,AA(1,LDAT1,MDAT1,1)
```

```
GOTO 10
```

C

C ***** 終了

```
20 CONTINUE
```

```
STOP
```

```
END
```

C

C

C データ入力部

```
SUBROUTINE SEIBI(IXM,IYM,IZM,EPA,VO,ZIN,INTIM,ICALT,IDTIM,
```

```
# AAK,ADK,AEK,VAA,VAD,LOADA,LOADD,JAA,JAD,JAЕ,MDV)
```

(以下は第4章にサンプルコードを示す。)

```
RETURN
```

```
END
```

3.2 周波数変換プログラム(SNMFREQ.EXE)

本プログラムでは、時系列で並んだデータをフーリエ変換により周波数データに変換する。フーリエ変換計算にはIMSL 10.0のサブルーチン、FFTRC、FFTCC、FFT2Cを使用している。

入力は、時系列に並べられたデータファイルから読み取る。データファイルは、1レコードに1ポイントもしくは複数ポイントのデータが記述され、そのレコードが時系列に並んだ構成とする。(図3-1)

図3-2は、計算結果の一例であって、解析対象に入力したときの入力ポートと出力ポートの観測結果を示している。入力信号データ[パルス波形]と反射波または透過波を周波数軸上で比を

入力ポートのデータ		出力ポートのデータ	
0.0	0.0	0.0	↓ 時 系 列
0.0100	0.00	0.00	
0.0523	0.03	0.03	
-----	-----	-----	
0.013	0.123	0.123	

図3-1 入力データ例




図3-2 計算データ例

取ればS11またはS21が求められる。計算精度をたかめるためには、さらに透過や反射の基準信号(たとえばスルーやショートの場合の波形)と比較下ほうがよい。

本プログラムではレファレンス信号として、入力信号データ[パルス波形]、透過信号データ[スルー波形]、反射信号データ[リターン波形]を扱うことができる。計算結果は指定周波数範囲のゲインと位相の値をファイルに出力する。

プログラムは対話形式で実行される。なお、以下に使用手順を簡単に示す。

使用手順

1 プログラムを起動する。表示が出て、計算パラメータ値(格子間隔[mm単位]、観測回数)を聞いてくる。以下のデータ処理において入力データ数がこの観測回数に到達しない場合は残りのデータを0として計算する。(ウインドはかけていない。)パラメータ値を入力すると計算条件が表示される。条件が良ければ、Yを入力する。

あなたの設定条件は以下のとおりです。

観測時間間隔 3.33333333[psec]

観測総時間 0.00000033[sec]

周波数分解能 3.00002994[MHz]

周波数分解能の設定条件はこれでよいですか?[Yes/No]

2 入力信号データ[パルス波形]のレファレンスデータについて問い合わせてくる。

入力信号データ[パルス波形]について

基準データを読み込みますか?[Yes/No]

データを入力する場合はYをインプットする。Yを入力するとデータファイル名(15文字以内)、タイプ(バイナリーは倍精度に限定)、項数(1レコードあたりの配列数)について問い合わせてくる。データファイルを読み込むとそのファイルにおけるレファレンスデータの位置(項数およびレコード番号の範囲)を問い合わせてくる。図4-2を例にとると、入力側の時刻T1までが入力波だから、レファレンスデータの項は1、レファレンスデータのレコード番号の範囲は1からT1までとなる。

3 同様に、透過および反射のレファレンスデータを順に問い合わせてくる。2と同様な操作を行う。なお、各ファイルを読み込んだあと次の表示がでる。

このデータの中に入力信号が含まれていますか?[Yes/No]

これは、読み取りデータから入力信号データ[パルス波形]のレファレンスデータを引き去る操作である。また、反射信号データの場合は、その基準条件(ショートかオープン)を問い合わせてくる。

4 つづいて解析対象データについて、2と同様に、データファイル名(15文字以内)、タイプ(バイナリーは倍精度に限定)、項数(1レコードあたりの配列数)について問い合わせてくる。データファイルを読み込むとそのファイルにおける解析対象データの位置(項数およびレコード番号の範囲)を問い合わせてくる。図4-2を例にとり、透過(S21)を計算するには、データの項は2、データのレコード番号の範囲は1からT2までとなる。計算が終わったら周波数変換作業を終了するか問い合わせてくる。Yが入力されるまでこの作業を繰り返す。

5 計算が終わったら出力データ設定としてデータファイル名(15文字以内)、タイプ(バイナリーは倍精度に限定)、出力周波数範囲を入力する。ファイルを作成してプログラムが終了する。

以下にプログラムコードを記す。

```

C *****
C                                     1988. 7. 25   by K. Tsunoda
C *****
C
C このプログラムは、空間回路網法で計算した任意形状回路の指定部分の電界値
C の時間軸観測データを用いて、これにFFTを用いて周波数軸における透過
C および反射特性を計算する機能を持つ。また読み込んだファイルを
C バイナナリーファイルとして出力することもできる。
C *****
C
C このプログラムでは IMSL サブルーチンパッケージの FFT ルーチンを
C 用いている。LINK の時に引用すること。
C *****
C 変数リスト
C VC:光速 [m/s]          PAI:円周率
C DELTA:ノード間距離 [mm]  DELS:観測時間間隔 [nsec]
C ITOTL:総観測回数      TTOTL:総観測時間 [sec]
C DELF:周波数分解能 [MHz]
C IFIL1:ファイル名 [15文字]  ITYP1:YES/NO 判別用文字ビット
C ICLM1:ファイルの項数      JREAD:読み取り作業位置認識用ビット
C ILIN1:ファイルの行数
C NO1:読み取り項数番号     NOS1,NOS2:データ収納行番号
C DATA1(*,*):ファイルからの読み取りデータ収納メモリー
C PIN(*):入力データ(時間軸) PDAT(*):観測データ(時間軸)
C *** 周波数軸データは、2行の配列であって、(1,*)は振幅 (2,*)は位相がはいる。
C EPIN(*,*):入力データ(周波数軸) ERL(*,*):反射基準データ(周波数軸)
C EDAT(*,*):観測データ(周波数軸)
C EYOUT(*,*):サブルーチン変数(周波数軸)
C KAISU:観測データ数
C FREQ:出力データの周波数値 [GHz] LU:出力LU No.
C
C *****
C      IMPLICIT DOUBLE PRECISION (D)
C      PARAMETER( MDATM=180000, MCLMM=15,ICALM=540450,MFFT=90001,
C # MCALM=60000)
C
C ** 配列宣言
C
C      DOUBLE PRECISION FREQ,DOUT(MCLMM)
C      COMPLEX CPIN(MDATM),CRL(MDATM),CPASS(MDATM),COUT(MDATM),
C # CC1,CC
C      DIMENSION DATA1(MCLMM,MCALM),PIN(MDATM),PDAT(MDATM),
C # EDAT(MCLMM,MFFT),JP(3)
C      CHARACTER*23 TIME
C      CHARACTER*15 IFIL1,IFILE,NAMEIN
C      CHARACTER*1 ITYP1,ITYP2
C
C      DATA PAI/3.1415927
C      DATA VC/2.997925E8
C
C      IFIL1='
C 101 FORMAT(A1)
C 115 FORMAT(A15)
C
C [1] ** 表示機能の設定
C

```

```

        WRITE(6,1101)
1101 FORMAT(' *****',/,
#       ' 空間回路網法で計算した電界値データを',/,
#       ' 用いて透過および反射特性を計算します。',/,
#       ' *****',/)
C
C #####
        WRITE(6,*)' 入力をファイルからしますか? YES=1 NO=0 '
        READ(5,*)JUDGE
        IF(JUDGE.EQ.1) THEN
            WRITE(6,*)' ファイル名を入力して下さい。'
            READ(5,115)NAMEIN
            IO=50
            OPEN(IO,FILE=NAMEIN,STATUS='OLD')
        ELSE
            IO=5
        ENDIF
C
C ***** 全体の計算時のパラメーター設定
C
        WRITE(6,1102)
1102 FORMAT(/,' *****',/,
#       '      計算に用いるパラメータ値の設定 ',/,
#       ' [ 計算に必要なパラメーターを入力してください ] ',/,
#       ' *****',/)
        WRITE(6,1105)
1105 FORMAT(/,' 隣接する同一ノード間隔はいくらですか ?[mm] ',/$)
        READ(IO,*)DELTA
        DELS=DELTA*1.0D9/VC/4.0
        5 WRITE(6,1106)
1106 FORMAT(/,' 総観測回数は何回にしますか [最大 180000 ] ?'./$)
        READ(IO,*)ITOTL
        I=ITOTL/2
        IF((2*I-ITOTL).NE.0)THEN
            ITOTL=ITOTL+1
            WRITE(6,*)' *** FFTのため観測回数を偶数回数にします。',ITOTL
        ENDIF
        TTOTL=(ITOTL-1)*DELS*1.0D-12
        DELF=1.0D-6/TTOTL
        WRITE(6,1107) DELS,TTOTL,DELF
1107 FORMAT(/,' あなたの設定条件は以下のとおりです。',/,
#       '      観測時間間隔 ',F15.8,'[psec]',/,
#       '      観測総時間 ',F15.8,'[sec]',/,
#       '      周波数分解能 ',F15.8,'[MHz]',/,/,
#       ' 周波数分解能の設定条件はこれでよいですか ? [Yes/No] ',/$)
        IF(JUDGE.EQ.0) THEN
            READ(IO,101)ITYP1
            IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 5
        ENDIF
C
C #####
C ***** 入力データの設定
C ***** データの読み取り及びデータの表示
C
        WRITE(6,1201)
1201 FORMAT(/,' *****',/,
#       ' 入力信号データ[パルス波形]について',/,

```

```

#           '   基準データを読み込みますか?[Yes/No]  ',/,
#           ' *****',/)
READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 501
JP(1)=1
JREAD=1
GOTO 90
C ***** データ範囲の設定 及び読み取り
10 CONTINUE
   DO I=1,NOS1-1
   PIN(I)=0.0
   ENDDO
   DO I=NOS1,NOS2
   PIN(I)=DATA1(N01,I)
   ENDDO
   DO I=NOS2+1,ITOTL
   PIN(I)=0.0
   ENDDO
C ***** 入力信号波形の FFT
   CALL ZFFT(PIN,CPIN,ITOTL,ND)
C
C ***** スルーデータ範囲の設定 及び読み取り
501 WRITE(6,1211)
1211 FORMAT(/,' *****',/,
#           ' 透過信号データ[スルー波形]について ',/,
#           '   基準データを読み込みますか?[Yes/No]  ',/,
#           ' *****',/)
READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 502
JP(2)=1
WRITE(6,1212)IFIL1
1212 FORMAT(' テータファイル[ ',A15,' ] に',/,
#           '   スルー基準データは含まれていますか?[Yes/No]',$,)
READ(IO,101)ITYP1
JREAD=2
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 90
GOTO 95
C
20 CONTINUE
   DO I=1,NOS1-1
   PDAT(I)=0.0
   ENDDO
   DO I=NOS1,NOS2
   PDAT(I)=DATA1(N01,I)
   ENDDO
   DO I=NOS2+1,ITOTL
   PDAT(I)=0.0
   ENDDO
C ***** 入力信号波形の FFT
   CALL ZFFT(PDAT,CPASS,ITOTL,ND)
C
C ***** 反射基準のデータ
C
502 WRITE(6,1221)
1221 FORMAT(/,' *****',/,
#           ' 反射信号データ[リターン波形]について ',/,
#           '   基準データを読み込みますか?[Yes/No]  ',/,
#           ' *****',/)

```

```

READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 503
JP(3)=1
WRITE(6,1222)IFIL1
1222 FORMAT(' テーザファイル[ ',A15,' ] に',/,
#      ' 反射基準データは含まれていますか?[Yes/No]',S)
READ(IO,101)ITYP1
JREAD=3
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 90
GOTO 95
30 CONTINUE
DO I=1,NOS1-1
PDAT(I)=0.0
ENDDO
DO I=NOS1,NOS2
PDAT(I)=DATA1(N01,I)
ENDDO
DO I=NOS2+1,ITOTL
PDAT(I)=0.0
ENDDO
WRITE(6,1223)
1223 FORMAT(' このデータの中に入力信号が含まれていますか?[Yes/No]')
READ(IO,101)ITYP1
IF((ITYP1.EQ.'Y').OR.(ITYP1.EQ.'y')) THEN
DO I=1,ITOTL
PDAT(I)=PDAT(I)-PIN(I)
ENDDO
ENDIF
C ***** 反射基準波形の FFT
CALL ZFFT(PDAT,CRL,ITOTL,ND)
WRITE(6,*) ' この反射データの基準は,ショート?オープン?'
WRITE(6,*) ' ショート基準 = 1,オープン基準 = 0'
READ(IO,*)JRL
C ***** 解析対象データの入力
503 CONTINUE
KAISU=0
48 WRITE(6,*) ' *****'
WRITE(6,*) ' 解析対象データについて'
WRITE(6,*) ' *****'
JREAD=4
GOTO 90
40 CONTINUE
42 GOTO 95
41 CONTINUE
IF(N01.EQ.0) GOTO 49
KKAIS=(KAISU+1)*3
IF(KKAIS.GT.MCLMM) THEN
WRITE(6,*) ' データ数がメモリーオーバーしていますので出力します'
GOTO 49
ENDIF
KAISU=KAISU+1
DO I=1,NOS1-1
PDAT(I)=0.0
ENDDO
DO I=NOS1,NOS2
PDAT(I)=DATA1(N01,I)
ENDDO
DO I=NOS2+1,ITOTL

```

```

PDAT(I)=0.0
ENDDO
WRITE(6,1223)
READ(IO,101)ITYP1
  IF((ITYP1.EQ.'Y').OR.(ITYP1.EQ.'y')) THEN
    DO I=1,ITOTL
      PDAT(I)=PDAT(I)-PIN(I)
    ENDDO
  ENDIF
C ***** 観測波形のFFT
  CALL ZFFT(PDAT,COUT,ITOTL,ND)
C ***** ゲイン計算
  I=KAISU*2-1
  WRITE(6,*)' このデータはどのような処理をしますか?'
  WRITE(6,*)'   サマエ-ク=1   '
  IF(JP(1).EQ.1) WRITE(6,*)' 入力波形との比=2'
  IF(JP(3).EQ.1) WRITE(6,*)' 反射波形との比=3'
  IF(JP(2).EQ.1) WRITE(6,*)' 透過波形との比=4'
  READ(IO,*)K
  DO J=1,ND
    IF(K.EQ.1) CC1=(1.0,0.0)
    IF(K.EQ.2) CC1=CPIN(J)
    IF(K.EQ.3) CC1=CRL(J)
    IF(K.EQ.4) CC1=CPASS(J)
    IF(ABS(CC1).NE.0) THEN
      CC=COUT(J)/CC1
      AR=REAL(CC)
      AI=AIMAG(CC)
      AA=SQRT(AR*AR+AI*AI)
C      WRITE(6,*)AA
      ELSE
      AA=0.0
      ENDIF
C
    IF(AA.GT.0.0) THEN
      EDAT(I,J)=20.0*LOG10(AA)
    ELSE
      EDAT(I,J)=0.0
    ENDIF
C
    IF(AR.NE.0.0) THEN
      AI=AI/AR
      IF((AI.LT.1.0E5).AND.(AI.GT.-1.0E5)) THEN
        EDAT(I+1,J)=ATAND(AI)
      ELSE
        IF(AI.GE.1.0E5) EDAT(I+1,J)=90.0
        IF(AI.LE.-1.0E5) EDAT(I+1,J)=-90.0
      ENDIF
      ELSE
      EDAT(I+1,J)=90.0
      ENDIF
C
    IF(AR.LT.0.0) THEN
      EDAT(I+1,J)=EDAT(I+1,J)+180.0
    ENDIF
    IF((JRL.EQ.1).AND.(K.EQ.3)) THEN
      EDAT(I+1,J)=180.0-EDAT(I+1,J)
    ENDIF

```

```

ENDDO
C
WRITE(6,1014)IFIL1,ICLM1,ILIN1
1014 FORMAT(' 入力データはつぎのとうりです。',/,
#'   ファイル名 : ',2X,A15,/, '   コラム数 : ',2X,I3,5X,
#'   データ行数 : ',2X,I6,/)
WRITE(6,*)' 周波数変換作業を終了しますか? [Yes/No]'
READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 50
GOTO 49
50 WRITE(6,*)' 別のファイルを使用しますか? [Yes/No]'
READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 42
GOTO 48
49 CONTINUE
C
C ***** データの出力
WRITE(6,*)' '
WRITE(6,*)' '
WRITE(6,*)' 出力データファイルを作成します。'
WRITE(6,*)' ファイルの名前を入力してください。[最大 15字]'
READ(IO,115)IFIL1
WRITE(6,*)' このデータは出力フォーマットを選択してください。'
WRITE(6,*)' バイナリーなら[1],アスキーなら[0]を入力。'
READ(IO,115)JOK
FREQ=DELF*(ND-1)
FREQ0=0.0
51 WRITE(6,*)' 現在計算したデータの周波数範囲はつぎのとうりです。'
WRITE(6,*)' 最大周波数',FREQ,'GHZ'
WRITE(6,*)' 最小周波数',FREQ0,'GHZ'
WRITE(6,*)' '
WRITE(6,*)' 出力データの周波数範囲を入力してください。[GHZ]'
READ(IO,*)FREQ0,FREQ1
FREQ2=MAX(FREQ0,FREQ1)
FREQ1=MIN(FREQ0,FREQ1)
WRITE(6,*)' 出力データの周波数範囲は OK ですね? [Yes/No]'
READ(IO,101)ITYP1
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) GOTO 51
IF(JOK.EQ.1)THEN
OPEN(LU,FILE=IFIL1,FORM='UNFORMATTED',STATUS='UNKNOWN')
DELF=DELF*1.0D-3
FREQ=-DELF
L=KAISU*2
ILIN1=0
DO I=1,ND
FREQ=FREQ+DELF
IF((FREQ.GE.FREQ1).AND.(FREQ.LE.FREQ2))THEN
ILIN1=ILIN1+1
DO J=1,L
DOUT(J)=EDAT(J,I)
ENDDO
WRITE(LU)FREQ,(DOUT(J),J=1..L)
ENDIF
ENDDO
ELSE
OPEN(LU,FILE=IFIL1,STATUS='UNKNOWN')
DELF=DELF*1.0D-3
FREQ=-DELF

```

```

L=KAISU*2
ILIN1=0
DO I=1,ND
  FREQ=FREQ+DELF
  IF((FREQ.GE.FREQ1).AND.(FREQ.LE.FREQ2))THEN
    ILIN1=ILIN1+1
    DO J=1,L
      DOUT(J)=EDAT(J,I)
    ENDDO
    WRITE(LU,*)FREQ,(DOUT(J),J=1,L)
  ENDIF
ENDDO
ENDIF
CLOSE(LU)
WRITE(6,1314)IFIL1,L,ILIN1
1314 FORMAT(' 出力データーはつぎのとうりです。',/,
#'   ファイル名:',2X,A15,/, '   コラム数:',2X,I3,5X,
#'   データ行数:',2X,I6,/, '*****')
STOP
C
C ***** データのファイルからの読み取り
98 WRITE(6,*) ' ファイルが在りません。再度 入力してください。'
  WRITE(6,*) ' ***** '
  WRITE(6,*) ' '
99 CONTINUE
  WRITE(6,*) ' '
  WRITE(6,*) ' そのファイルの名前を入力してください。[最大 15字]'
  READ(IO,115)IFIL1
  WRITE(6,*) ' それはバイナリーファイルですか? [Yes/No]'
  READ(IO,101)ITYP1
  WRITE(6,*) ' そのファイルの項数はいくつですか?'
  READ(IO,*)ICLM1
  WRITE(6,*) ' *** データ読み取り中 ***'
  LU=70
  ILIN1=0
  IF((ITYP1.EQ.'Y').OR.(ITYP1.EQ.'y')) THEN
    OPEN(LU,FILE=IFIL1,FORM='UNFORMATTED',ERR=98,STATUS='OLD')
    DO ILIN1=1,MDATM
      READ(LU,END=91)(DATA1(J,ILIN1),J=1,ICLM1)
    ENDDO
91  CONTINUE
    CLOSE(LU)
    ILIN1=ILIN1-1
  ELSE
    OPEN(LU,FILE=IFIL1,ERR=98,STATUS='OLD')
    DO ILIN1=1,MDATM
      READ(LU,*,END=92)(DATA1(J,ILIN1),J=1,ICLM1)
    ENDDO
92  CONTINUE
    CLOSE(LU)
    ILIN1=ILIN1-1
  ENDIF
C
C ***** バイナリーファイルの出カルーチン
IF((ITYP1.EQ.'N').OR.(ITYP1.EQ.'n')) THEN
  WRITE(6,*) ' バイナリーファイルとして出力しますか? [Yes/No]'
  READ(IO,101)ITYP1
  IF((ITYP1.EQ.'Y').OR.(ITYP1.EQ.'y')) THEN

```

```

        IFILE='
        WRITE(6,*)'出力ファイル名を入力してください。'
        READ(IO,115)IFILE
        OPEN(LU,FILE=IFILE,FORM='UNFORMATTED',STATUS='UNKNOWN')
        DO I=1,ILIN1
            WRITE(LU)(DATA1(J,I),J=1,ICLM1)
        ENDDO
        CLOSE(LU)
        WRITE(6,*)'記入を終わりました。'
    ENDIF
    ENDIF
    IF(JREAD.EQ.4) GOTO 40
C
    95  WRITE(6,1014)IFIL1,ICLM1,ILIN1
        WRITE(6,1202)
    1202 FORMAT(/,'そのデータは、第何番目の項に入っていますか?',/,
#          '          [データがない時は0を入力してください。]','$)
        READ(IO,*)NO1
            IF(NO1.GT.ICLM1) THEN
                WRITE(6,*)'指定された項はこのデータファイルには在りません。'
                GOTO 95
            ENDIF
        IF(NO1.GT.0) THEN
            WRITE(6,1203)
    1203 FORMAT(/,'そのデータは、第何番目から第何番目にありますか?','$)
            READ(IO,*)NOSS1,NOSS2
            NOS1=MIN(NOSS1,NOSS2)
            NOS2=MAX(NOSS1,NOSS2)
        ENDIF
        GOTO (10,20,30,41) JREAD
        END
C
C ***** 実時間データをもとに FFT を用いて周波数ドメインへの変換を行う
        SUBROUTINE ZFFT(EYIN,FIN,ITOTL,ND)
C *****
C 変数リスト
C MDATM: データの最大数  ICALM: FFT用メモリーのサイズ
C  EYIN(MDATM) : 時間軸データ  EYOUT(2,MFFT) : 周波数軸データ
C  LINM : 時間軸データ数  WK(ICALM),IWK(ICALM) : FFT 計算用ワーク
C  FIN(MFFT) : FFT 変換複素係数
C *****
        PARAMETER( MDATM=180000, MCLMM=10,ICALM=540450,MFFT=90001,
# MCALM=60000)
        COMPLEX FIN(MDATM)
        DIMENSION EYIN(MDATM),IWK(ICALM),WK(ICALM),EYOUT(2,MFFT)
C ***** 定数設定
        WRITE(6,*)' *** FFT による周波数軸変換を行っています。'
C ***** データ入力ファイルの指定およびデータの読み込み
        ND=ITOTL/2+1
C ***** FFTルーチン
        CALL FFTRC(EYIN,ITOTL,FIN,IWK,WK)
        WRITE(6,*)' *** FFT 計算が終わりました。'
C ***** 出力
        RETURN
        END

```

4. 使用例

前記プログラムの使用例を以下に記す。

4.1 マイクロストリップ共振回路特性の計算

図4-1に示す マイクロストリップ共振回路特性の計算例を以下に示す。周波数特性を計算するために、スルーとショートを基準に用いた。

4.1.1 サブルーチンの作成

例題においては、最低間隔が0.636mmであり、以下の計算においては、 $\Delta d = 0.212\text{mm}$ の格子点を考える。また、Oコーナを座標基準点として、各ノードの座標値を設定する。図4-2に各要素の座標値を示す。

誘電体は厚み0.636mm、すなわち $3\Delta d$ である。よって、D・Eのノードは $y = 1 \sim 3$ のものが誘電媒質、 $y = 4$ のものが誘電体・空気境界面、他は空気となる。一方、Aノードは $y = 1 \sim 3$ のものが誘電媒質、他は空気となる。

金属はX-Z平面にのみ設定されている。まず、アース面として、D・Eノードの $y = 1$ に金属条件を設定する。ストリップ導体は、D・Eノードの $y = 4$ の面上にある。図よりDノードは $x = 23 \sim 25$, $z = 61 \sim 121$ 、 $x = 29 \sim 31$, $z = 31 \sim 91$ 、 $x = 35 \sim 37$, $z = 1 \sim 61$ 、Eノードは $x = 23 \sim 26$, $z = 61 \sim 120$ 、 $x = 29 \sim 32$, $z = 31 \sim 90$ 、 $x = 35 \sim 38$, $z = 1 \sim 60$ が金属となる。

励振は、 $z = 1$ 面において $y = 3, x = 36, 37$ のAノードに設定する。受信は、 $z = 121$ 面において $y = 3, x = 24$ の点とした。なお、反射波の観測点は励振点とした。

励振はレイズドコサインパルス(パルス幅 $80\Delta t$)、観測は $20000\Delta t$ とした。

上記モデルにより作成したサブルーチンを以下に示す。

```
C
C   データ入力部
      SUBROUTINE SEIBI(IXM,IYM,IZM,EPA,VO,ZIN,INTIM,ICALT,IDTIM,
#   AAK,ADK,AEK,VAA,VAD,LOADA,LOADD,JAA,JAD,JAЕ,MDV)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
C ***** 配列宣言
      DIMENSION
#   AAK(IXM,IYM,IZM),ADK(IXM,IYM,IZM),AEK(IXM,IYM,IZM),
#   JAA(IXM,IYM,IZM),JAD(IXM,IYM,IZM),JAE(IXM,IYM,IZM),
#   VAA(IXM,IYM),VAD(IXM,IYM),LOADA(IXM,IYM),LOADD(IXM,IYM)
C
      LV=IXM
      MV=IYM
      NV=IZM
C ***** データの読み込み
C   INTIM : 入力電源のレイズドコサイン周期   ICALT : 計算回数   IDTIM :
      INTIM=80
      ICALT=20000
      IDTIM=80
C   EPA   : 誘電率   VO   : 電源電圧   ZIN   : 入力インピーダンス
      EPA= 10.0
      VO= 1.0
      ZIN= 50.0
C   MDV: 誘電体層の厚み
      MDV=4
C ***** 誘電体条件
C   1: 空気   2: 誘電体境界面   3: 誘電体
C   初期値 空気
      DO 10 N=1,NV
      DO 11 M=1,MDV-1
```

DO 11 L=1, LV
 JAA(L, M, N)=3
 JAD(L, M, N)=3




図4-1 マイクロストリップ共振回路




図4-2 ノード座標




図4-3 電界分布の時間変化




図4-4 周波数特性計算値

```

11   JAE(L,M,N)=3
      DO 12 L=1,LV
          JAA(L,MDV,N)=1
          JAD(L,MDV,N)=2
12   JAE(L,MDV,N)=2
      DO 13 M=MDV+1,MV
          DO 13 L=1,LV
              JAA(L,M,N)=1
              JAD(L,M,N)=1
13   JAE(L,M,N)=1
10   CONTINUE
C ***** 電氣的ノードの設定条件設定
C           0   金属           1  拘束条件なし
      DO 20 N=1,NV
          DO 20 M=1,MV
              DO 20 L=1,LV
                  AAK(L,M,N)=1.0
                  ADK(L,M,N)=1.0
20   AEK(L,M,N)=1.0
C   アース
      DO 21 N=1,NV
          DO 21 L=1,LV
              ADK(L,1,N)=0.0
21   AEK(L,1,N)=0.0
C   ストリップ導体
      DO 22 I=1,61
          DO 22 J=1,3
              ADK(22+J,MDV,60+I)=0.0
              ADK(28+J,MDV,30+I)=0.0
22   ADK(34+J,MDV,I)=0.0
      DO 23 I=1,60
          DO J=1,4
              AEK(22+J,MDV,60+I)=0.0
              AEK(28+J,MDV,30+I)=0.0
              AEK(34+J,MDV,I)=0.0
          ENDDO
      ENDDO
C
C ***** Z=0 における電源設定ノードの条件入力
      DO M=1,MV
          DO L=1,LV
              VAA(L,M)=0
              VAD(L,M)=0
          ENDDO
      ENDDO
      VAA(36,3)=1
      VAA(37,3)=1
C
      RETURN
      END

```

4.1.2 出力部の修正

出力は、Y方向電界値を取ることにする。また、一定回数ごとのY方向電界分布を観測するために配列を出力する。よって、次の様になる。

```

C *****出力部
C ***** エータ アウトプット
C ***** 電界強度データ出力用

```

```

WRITE(60,*)IT,AA(1,11,2,1),AA(1,6,2,21)
ITT=ITT+1
IF(ITT.EQ.20)THEN
DO J=1,MV
WRITE(60,*)(AA(1,I,J,3),I=1,LV)
ENDDO
ITT=0
ENDIF
GOTO 10

```

4.1.3 計算結果

以上のサブルーチンを用いてプログラムを実行すると、図4-3にしめすような電磁界伝搬の時間列データが得られる。図より入力パルスがストリップ線路に沿って進み、横結合部分で隣のストリップ線路が励振されていくところが観測できる。

入力および出力ポートにおける電界値の時間列データを用いて周波数特性を計算すると図4-4となる。基本波、2倍波を含め広帯域な特性が得られる。

4.2 開放領域の取扱い

本解析においては、自由空間は吸収壁として表現される。よって、放射の大きな素子を解析する場合は、解析領域を充分大きく取らなければならない。図4-5に板状逆Fアンテナの入カインピーダンスの計算結果をしめす。図は解析領域を変化したときの入力インピーダンスの軌跡を示している。図より分かるように解析領域が小さい場合は自由空間境界が電波吸収体のように働き、等価的にQが小さくなることがわかる。充分大きくとれば計算値は実測値とよく一致する。

このように、開放領域を必要とするものは解析領域の設定に注意を払わなければならない。




図4-5 アンテナ特性解析例

5. まとめ

電磁界の時間軸解析手法のひとつである空間回路網法について解析法の概略とプログラム作成上必要となる計算式について記した。さらに本解析法のプログラムのコードおよび使用法を示し、解析を行う上での注意点について記した。

本手法は近年、電子通信情報学会等において多くの報告がなされている。それに伴い、不等間隔格子点の導入、斜め入射波に対する吸収境界条件の設定法など本手法の欠点克服のための改良研究も盛んに報告されている。

今後この解析法に関する報告を見る機会もあると思うが、そのときにこのレポートが参考になれば幸いである。

改訂記録等

1990年7月27日初版作成

***** プログラム使用者へのお願い

このレポートは、FUJI XEROX J Star で作成しており、文書ファイルはフロッピーに納めています。本文章の誤り、プログラムの修正・改良等変更事項があった場合は、改訂記録を追加するとともに本文の修正も行って下さい。